```
/*Evaluation of FLuorometholone as Adjunctive MEdical Therapy for Trachomatous Trichiasis Surgery: The FLAME Randomized
Controlled Clinical Trial*/
/*purpose: Comparison of patient-reported outcomes between randomized treatment groups*/
/*Note: raw datasets in SAS library 'in' corresponds to the shared data file in xlsx format, their relationship are as below:
in.satisfaction_surgery_m12 - month 12 record in SS
in.SURGERY - SI
in.EQ5D_M12 - month 12 record in EQ
in.exit - EX*/

/*set up SAS library, TLF path, TLF title, format catalog which will be used in later analysis*/
%include "setup.sas";
libname in "&_root_in";
libname fm "&_root_fm";
libname data "&_root_data";
%let TLF_path = &_root_output\tb6_satisfaction.rtf;
%let TLF_title = Table 6:  Comparison of patient-reported outcomes between randomized treatment groups;

OPTIONS FMTSEARCH=(fm.fmsurgery fm.fmsat fm.fmeq fm.fmexamw4 fm.fmelig);
option mprint;

/*set up variable format for visualization*/
proc format;
        value gf  0 = "Placebo" 1 = "FML";
        value ynf 0 = "No" 1 = "Yes";
        value yno 1='No' 2='Yes';
        value satf 0 = "Very Satisfied"
                   1 = "Satisfied"
```

```
                    2="Neither Satisfied nor dissatisfied"
                    3="Dissatisfied"
                    4 = "Very dissatisfied";
        value sato 1 = "Very Satisfied"
                    2 = "Satisfied"
                    3="Neither Satisfied nor dissatisfied"
                    4="Dissatisfied"
                    5 = "Very dissatisfied";
        value pval (default=8)
                low - <0.00095 = '<0.001'
                0.00095 - <0.0095 = [8.3]
                0.0095 - <0.045 = [8.2]
                0.045 - <0.0495 = [8.3]
                0.0495 - <0.04995 = [8.4]
                0.04995 - <0.05 = '~<0.05'
                0.05 = '0.05'
                0.05< - <0.05005 = '~>0.05'
                0.05005 - <0.0505 = [8.4]
                0.0505 - <0.055 = [8.3]
                0.055 - 0.99 = [8.2]
                0.99< - high = [8.2];
run;

/*combine covariate for analysis*/
data SATISFACTION_SURGERY_M12;
        set in.satisfaction_surgery_m12;
run;
```

```
%proc_reshape(indata=WORK.SATISFACTION_SURGERY_M12,memname="SATISFACTION_SURGERY_M12",outdata=SATISF
ACTION_SURGERY_M12_);

data rand;
	set data.rand;
	if reye_eligibel=1 then do; eye='OD'; output; end;
	if leye_eligibel=1 then do; eye='OS'; output; end;
run;

proc sql;
	create table data.tbl6_data_1 as
	select r.subjid, r.eye, r.treat,
	a.sstrich, a.sscosmetic,
	case when a.sstrich=0 then 0
		when a.sstrich in (1,2,3,4) then 1 end as sstrichc,
	case when a.sscosmetic=0 then 0
		when a.sscosmetic in (1,2,3,4) then 1 end as sscosmeticc,
	/*to meet requirement of HIPAA, name (sisiname2) and certification (sisicert2) of surgeon is removed from the public
shared dataset*/
	sisicert2,sisiname2,
	/*for surgeon that only have a small number of surgery (<50), cluster them into 1 group*/
	case when sisicert2 in (/*surgeon license removed*/) then 'Other'
		else sisicert2 end as sisicert3
from rand as r
inner join SATISFACTION_SURGERY_M12_ as a
on r.subjid=a.subjid and r.eye=a.eye
left join in.surgery as s on r.subjid=s.subjid;
```

```
quit;

proc sql;
	create table data.tbl6_data_2 as
	select r.subjid, r.treat,
	b.eqmobility + b.eqselfcare + b.eqactivities + b.eqpain + b.eqanxiety as Sev_index,
	b.eqhealthscale,
	c.exscale,
	/*to meet requirement of HIPAA, name (sisiname2) and certification (sisicert2) of surgeon is removed from the public
shared dataset*/
	sisicert2,sisiname2,
	/*for surgeon that only have a small number of surgery (<50), cluster them into 1 group*/
	case when sisicert2 in (/*surgeon license removed*/) then 'Other'
		else sisicert2 end as sisicert3
from data.rand as r
left join in.EQ5D_M12 as b on r.subjid=b.subjid
left join in.exit(where=(ex12month=1)) as c on r.subjid=c.subjid
left join in.surgery as s on r.subjid=s.subjid;
quit;

/*generate number of eyes by group for header*/
proc sql noprint;
select count(*) into :t0_eye
		from data.tbl6_data_1
		where treat = 0;
select count(*) into :t1_eye
		from data.tbl6_data_1
```

```
                where treat = 1;
quit;
%put &t0_eye;
%put &t1_eye;

*descriptive statistics, hypothesis test and model building;
%stat_char(input=data.tbl6_data_1,output=tab_1_1,unit=eye,variable=sstrich,min=1,max=5,cohort=treat,ref=Placebo,questi
onlabel=Degree of satisfaction with the trichiasis surgery outcome in study eyes at 1
year,tab_order=1,value_label=satf,order_label=sato,id=subjid,gee_binomial=0,gee_multinomial=1,reverse=0,chi=0,fisher=0,s
urgeon=1,surgeon_cluster=1,surg_bar=0);
%stat_char(input=data.tbl6_data_1,output=tab_1_2,unit=eye,variable=sstrichc,min=1,max=1,cohort=treat,ref=Placebo,quest
ionlabel=,tab_order=1,value_label=ynf,order_label=yno,id=subjid,gee_binomial=1,surgeon=1,surgeon_cluster=1,surg_bar=0,d
iffs=1,event=0);
data tab_1;
        merge tab_1_1 tab_1_2(keep=answer_order diffs);
        by answer_order;
run;
proc datasets;
        delete tab_1_1 tab_1_2;
quit;
%stat_char(input=data.tbl6_data_1,output=tab_2_1,unit=eye,variable=sscosmetic,min=1,max=5,cohort=treat,ref=Placebo,q
uestionlabel=Degree of satisfaction with the cosmetic outcome in study eyes at 1
year,tab_order=2,value_label=satf,order_label=sato,id=subjid,gee_binomial=0,gee_multinomial=1,reverse=0,chi=0,fisher=0,s
urgeon=1,surgeon_cluster=0,surg_bar=0);
%stat_char(input=data.tbl6_data_1,output=tab_2_2,unit=eye,variable=sscosmeticc,min=1,max=1,cohort=treat,ref=Placebo,
questionlabel=,tab_order=2,value_label=ynf,order_label=yno,id=subjid,gee_binomial=1,surgeon=1,surgeon_cluster=0,surg_b
ar=0,diffs=1,event=0);
```

```
data tab_2;
	merge tab_2_1 tab_2_2(keep=answer_order diffs);
	by answer_order;
run;
proc datasets;
	delete tab_2_1 tab_2_2;
quit;

%stat_num(input=data.tbl6_data_2,output=tab_3,variable=exscale,cohort=treat,ref=Placebo,questionlabel=,tab_order=3,linear=1,surgeon=1,diffs=1);
%stat_num(input=data.tbl6_data_2,output=tab_4,variable=Sev_index,cohort=treat,ref=Placebo,questionlabel=Health Utility by EQ-5D at 1 year**,tab_order=4,linear=1,surgeon=1,diffs=1);
%stat_num(input=data.tbl6_data_2,output=tab_5,variable=eqhealthscale,cohort=treat,ref=Placebo,questionlabel=,tab_order=5,linear=1,surgeon=1,diffs=1);

data data.tbl6_result;
	length answerlabel diffs $200;
	set tab_:;
	if tab_order not in (3,4,5) or answer_order in (2,0);
	if tab_order=3 then answerlabel='Scale of satisfaction with the outcome of TT surgery at 1 year*: Mean (SD)';
	if tab_order=4 and index(answerlabel,'Mean (SD)')>0 then answerlabel='  Severity index (0-25, higher is worse): Mean (SD)';
	if tab_order=5 then do; answerlabel='  Health state score (0-100, higher is better): Mean (SD)'; tab_order=4; answer_order=3; end;
run;

/*table visualization*/
```

```
options nodate nonumber;
ods rtf file = "&TLF_path";
ods escapechar='~';
ods rtf text="~S={just=l font_size=10pt font_weight= bold} &TLF_title";
proc report data=data.tbl6_result nowd spanrows split='|' missing style(column)={background=white fontsize=9pt}

style(header)={background=white fontsize=9pt fontweight=medium};
        columns tab_order answer_order answerlabel cohort_0 cohort_1 diffs pvalue;
        define tab_order  / ' ' order order=internal noprint;
        define answer_order  / ' ' order order=internal noprint;
        define answerlabel / 'Patient-reported outcomes' left style(column)={cellwidth=1.8in asis=ON} style(header)={just=l};
        define cohort_0 / "Placebo|(n=%trim(&t0_eye) study eyes)" center style(column)={cellwidth=1.4in};
        define cohort_1 / "FLuorometholone|(n=%trim(&t1_eye) study eyes)" center style(column)={cellwidth=1.4in};
        define diffs / 'Difference (95% CI)' center group style(column)={cellwidth=1.4in};
        define pvalue / 'P-value#' center group style(column)={cellwidth=1.4in};
        compute after tab_order;
                line '' ;
         endcomp ;
run ;
ods rtf text="~S={just=l font_size=9pt} *On the scale of 1 to 10 (higher score indicate more satisfaction).";
ods rtf text="~S={just=l font_size=9pt} **Measured at participant-level instead of eye-level.";
ods rtf text="~S={just=l font_size=9pt} #For eye-level variables, derived from generalized regression models adjusting for
surgeon that account for inter-eye correlation; for patient-level variables, derived from linear regression adjusting for
surgeon.";
ods rtf close;
```